韓國數學敎育學會 주최 제 11회
國際數學英才敎育세미나 프로시딩
2006. 2. 25-37

Proceedings of the 11th International Seminar on
Education of Gifted Students in Mathematics
February 2006. 25-37

수학영재교육에서 스프레드 쉬트의 활용[1)]

Deane Arganbright (KAIST)

영재를 위한 수학교육은 우리의 당면과제 중 하나이다. 능력 있는 학생들의 학습이 속진에 한정되는 것보다는 심화자료 및 수학적 소프트웨어와 함께 하는 것이 더 의미 있을 것으로 기대된다.
본 연구는 스프레트쉬트를 사용한 수학적 아이디어의 탐구에 관한 것이다. 다음에 대해 논의하기로 하겠다.
i) 스프레드쉬트는 비전통적이면서도 이용이 용이하며, 수학적 통찰을 위한 매개물이다.
ii) 풍부하고, 흥미롭고, 가치있는 수학적 주제에 대해 스프레드쉬트를 이용할 수 있다.
iii) 스프레드쉬트를 사용하여 학생들이 수학적 아이디어에 대한 흥미를 고취시킬 수 있다.
iv) 스프레드쉬트는 학생들에게 그들의 창의적인 시각화 기술을 공개할 기회를 줌으로써 수학에 대한 폭넓은 도식적 이해를 제공한다.
v) animation을 포함한 스프레드쉬트 도식들의 적절한 사용은 유익하면서도 흥미롭다.
vi) 학생들은 일상생활에 나타나는 수학의 흥미로움을 발견할 것이다.
vii) 교사는 지금의 지도방식에 스프레드쉬트를 통합할 수 있다.

특히 스프레드쉬트는 다음과 같은 면모도 가지고 있다.
i) 창의적인 수학적 스프레드쉬트 모델들의 실제 과정들이 그 자체로써 수학적 개념발달에 이용될수 있다.
ii) 스프레드쉬트 모델은 심화된 주제의 탐색을 위한 의미 있는 탐구과제를 제공한다.
iii) 스프레드쉬트는 현장에서 사용되는 실제적 수학 도구이다. - 과학자나 공학도들의 사용도 증가되고 있다. 이것의 사용은 학생들이 현장에서 사용할 기술을 취득하게 할 수 있고, 같은 컴퓨터의 소프트웨어를 사용하는 가족의 대화 수단이 되기도 한다.

본 연구에서 우리는 스프레드쉬트의 4가지 실증적 예를 들어 보겠다.
또한 다른 영역에서 발전된 스프레드쉬트 모델의 몇 가지 도식적 산출물도 포함 할 것이다.
우리는 가장 대중적인 스프레드 쉬트인 Microsoft Excel 프로그램을 사용하였다. Excel의 수행과 Excel 연산의 설명을 담은 CD와 함께 다양한 사례들에 대한 논의는 [8]을 참고하기 바란다. 본고에서는 graphic animation 기술, 스크롤바의 사용을 간단하게 개괄하겠다. '동적형상들(movies)'를 만들 수 있는 간단한 매크로의 사용 등의 내용들은 각 자료를 사용할수 있는 Excel 파일의 예와 함께 [1]과 [8]에 설명하였었다. 많은 인쇄물과 on-line 참고문헌, 매체자료들도 함께 제공하였다.

1) 이 논문은 한국수학교육학회지 시리즈 D <수학교육연구> 제10권 1호(통권 25호)에 게재된 논문인 Using Spreadsheets with Mathematically Gifted Students를 번역한 것입니다.

I. 기하학

기하 영역은 종종 수학적으로 흥미롭고, 도전적이고 이해하기 쉬운 다양한 주제로 수학적 아이디어와 기술을 탐구하기 위해 소개된다. 여기서 우리는 스프레드쉬트의 동적 기술을 시연하는 예를 소개하겠다.

스프레드쉬트를 수학에서 사용하는 두 가지 일반적인 접근법이 있는데, 하나는 일반적인 방법으로 수학을 유도하고, 그래서 이것을 설명하는 동적 스프레드쉬트 모델을 설계하고 만드는 것이다. 다른 하나는 스프레트쉬트의 생성 과정자체를 수학적 발달을 위해 이용하는 것이다. 일례로 우리는 타원이 어떻게 만들어지는지 보기 위해 두점에 끈을 연결하고 이 끈의 한 점에서 처음 고정시킨 두 점에 이르는 거리의 합이 같은 점들의 자취를 탐구할 수 있는데, 이것에는 처음의 것을 적용할 것이고, 두 번째 접근법으로는 몇몇 사례들의 산출들을 알아볼 것이다.

초점이 $(-c,0)$, $(c,0)$인 타원을 만들기로 하자. 이 점들에 끈을 고정시킨다. fig1의 예에서 우리는 $c=4$로 하고, 끈의 길이를 $2a=10$로 하였다. 우리는 타원의 장축이 $a=5$, 단축이 $b=\sqrt{a^2-c^2}=3$임을 보이기 위해서 표준편차(standard deviation)을 사용할 수 있다. 결론적으로 curve는 매개방정식 $x=a\cos t$, $y=b\sin t$ (for $o\le t\le 2\pi$)로부터 그려진다. 우리의 모델은 타원의 동적 자취를 만들기 위해 스크롤 바를 통합하여 curve를 이끌어 낼 것이다. 우리의 초기 모델은 <그림 1>에, 수치적 결과는 <그림 2>에 제시되어 있다.



<그림 1>

	A	B	C	D	E	F	G	H
1	a	5	b	3	c	4		
2	N	4	x	4.988	y	0.209		
3								
4	k	t	x	y	X	Y	xx	yy
5	0	0	5	0	5	0	4	0
6	1	0.017	4.999	0.052	4.999	0.052	4.988	0.209
7	2	0.035	4.997	0.105	4.997	0.105	-4	0
8	3	0.052	4.993	0.157	4.993	0.157		
9	4	0.07	4.988	0.209	4.988	0.209		
10	5	0.087	4.981	0.261	4.988	0.209		

<그림 2>

<그림 3>에서 B1, D1 셀에 a, b 변수를 입력하면서 시작하고 있다. 그러면 F1 셀에는 c의 값으로 $\sqrt{a^2-b^2}$이 계산되도록 한다. 또는 a, c값을 입력하고 b 값으로 $\sqrt{a^2-c^2}$를 얻을 수 있다. 다음은 0도에서 360도까지 1을 단위로 하는 각도(degree) k를 A열에 만들고, *Excel* 라디안 함수를 사용해서 B열에 라디안 t가 출력되도록 한다. C 와 D열에는 $a\cos t$ 와 $b\sin t$이 출력되도록 한다. 변수 a와 b의 참조는 절대참조(ex. $B$2와 같이 함. not B2.)이고, t는 상대참조이다. 이 모든 단계에서 우리는 C:D열을 마우스를 사용하여 선택한 다음, 차트마법사의 분산형 그래프를 사용하여 타원그래프를 얻을 수 있다.

이제 애니메이션 효과를 만들어 보자. 그래서 셀 B2에 계수기 N의 값을 담는다. 여기서 N은 $0 \le N \le 360$인 정수이다. <그림 2>는 $N=4$일 때의 결과를 보여준다. E열과 F열에 *엑셀* 내장함수인 IF를 이용하여 $k \le N$이면 (x, y)의 현재 값을 제공하고, 그렇지 않으면 바로 앞셀의 값을 출력하게 하자. 이렇게 하면 $0 \le k \le N$ 범위의 점들만 그리게 하는 효과가 있다. 우리는 스크롤바를 이용하여 N의 값을 1씩 반복해서 증가시킬 수 있으며, 이러면 점이 추가로 만들어져서 원하는 곡선을 그리게 된다.

	A	B	C	D	E	F	G	H
1	a	5	b	3	c	=SQRT(B1^2-D1^2)		
2	N	4	x	=VLOOKUP(B2,t,3)	y	=VLOOKUP(B2,t,4)		
3								
4	k	t	x	y	X	Y	xx	yy
5	0	=RADIANS(A5)	=$B$1*COS(B5)	=$D$1*SIN(B5)	=IF($A5<=$B$2,C5,E4)	=IF($A5<=$B$2,D5,F4)	=F1	0
6	=1+A5	=RADIANS(A6)	=$B$1*COS(B6)	=$D$1*SIN(B6)	=IF($A6<=$B$2,C6,E5)	=IF($A6<=$B$2,D6,F5)	=D2	=F2
7	=1+A6	=RADIANS(A7)	=$B$1*COS(B7)	=$D$1*SIN(B7)	=IF($A7<=$B$2,C7,E6)	=IF($A7<=$B$2,D7,F6)	=-G5	0

<그림 3>

다음은 스크롤바를 만드는 방법이다. 메인 명령 메뉴에서 '보기> 도구모음> 컨트롤 도구 상자'를 차례로 선택한다. 이러면 도구상자 하나가 만들어지는데, 이것의 왼쪽 위 버튼을 누르면 '디자인 모드'로 들어간다. 이제 스크롤바 아이콘을 클릭한다.

<그림 4>

다음으로, 우리는 마우스를 드레그하여 스크롤바를 만들고 오른쪽 클릭을 이용해서 '연결된 셀' 칸에는 B2셀을, '최대값'칸은 360으로 입력한다. 이제 대화창을 닫고 디자인 모드를 벗어나면 우리가 만든 스크롤바는 B2셀에 연결되었다. 바(bar)를 오른쪽으로 움직이면, N의 값은 증가하여 그래프 상에 점을 추가로 만들며, 학생들은 그 점들에 의해 곡선이 형성되는 것을 관찰할 수 있다.

결과물을 더욱 깔끔하게 만들고 싶다면, VLOOKUP함수를 이용하여 현재 n값이 가리키는 점의 좌표를 생성하도록 한다. 그 다음 G:H 열에 새로운 수열을 우리의 그래프와 병합하여 점 $(-k, 0)$, (x, y), $(k, 0)$을 나타낸다. 또 타원 그래프를 그리는 끈과 연필 등의 도구를 나타낼 수 있는데, fig 5와 같이 연필 끝이 그래프를 그리는 이미지를 사용하여 시각적 산출물을 더욱 개선할 수 있다. 스크롤바의 슬라이드단추가 움직이면 연필도 따라 움직인다. 연필끝에서 각 초점에 이르는 거리는 $(-k, 0)$, (x, y), $(k, 0)$으로부터 계산하여 <그림 5>와 같이 나타낼 수 있다.



<그림 5>


<그림 6>

<그림 6>은 타원을 만드는 다른 예시이다. 이것은 방정식으로부터 얻어진 것이 아니라, 실제적인 스프레드시트 생성과정을 통해 얻어진 것이다. 이것을 만드는 것은 연습문제로 남겨둔다. 이것과 관련된 기계장치는 *trammel*이다. *trammel*은 단단한 막대기인데, x-축, y-축을 따라 나 있는 홈을 따라 움직이는 두 점에는 못이 달려 있으며, 막대기의 한쪽 끝에는 연필이 달려있다. 홈을 따라 핀들을 움직이면 연필은 우리가 그것이 타원임을 증명할 수 있는 곡선을 그리게 된다. 학생들은 책을 통해, 혹은 웹서핑을 통해, 이와 비슷한 특징을 가진 여러 가지의 장치들을 발견하여 탐구할 수 있을 것이다.

우리는 동일한 스프레드시트 접근법을 이용하여 극형식으로 주어진 다양한 매개방정식의 그래프를 그릴 수 있다. fig 7는 파리의 여행에 관한 것으로 $x=\cos 3t$, $y=\sin 5t$로 주어진 리사주 곡선을 그리는 중이다. 이때 우리는 선형대수의 지식을 통해 파리의 상을 회전시키고, 확대/축소하고, 평행이동하여, 그것이 언제나 주어진 곡선에 접하도록 만든다. 이 작도의 구체적인 방법은 [1]에서 온라인으로 확인할 수 있다. 우리는 또한 극좌표의 방정식에 대해, 먼저 $r=f(t)$를 계산한 다음, 식 $x=r\cos t$, $y=r\sin t$을 통해 $x-$좌표와 $y-$좌표를 구함으로써, 그것을 그래프로 그릴 수 있다. fig 8은 $r=\cos 2t$의 그래프에 대한 것이다.



<그림 7>



<그림 8>



<그림 9>

영재학생들을 위한 또 하나의 훌륭한 프로젝트는 전통적인 기하학적 작도에 등장하는 그림-예컨대 fig 9의 페달곡선-을 만드는 것이다. 우리는 파란색 델토이드deltoid을 따라 움직이는 점에서의 접선, 그리고 이 직선과 수직이며 주어진 고정된 점을 지나는 직선을 그릴 수 있다. 이 두 직선의 교차점이 움직이면서 페달곡선을 만들어낸다. 슬라이드 막대를 이용하여 그 점을 델토이드를 따라 움직이도록 하면, 페달이 그려지는 것을 볼 수 있다.

II. 함수 탐구하기

스프레드시트 그래픽은 학생들에게 함수의 움직임에 대한 시각적 연구를 할 수 있는 무한한 능력을 제공한다. 게다가, 일반적으로는 그래프 계산기로 할 수 없는 방법으로 함수를 만들고 분석할 수 있다. 또한 학생들은 독창적인 방법으로 산출물을 움직여 볼 수 있다. 여기에서 우리는 이용할 수 있는 무수히 많은 개념들 중 하나만, 즉 함수의 수평 평행이동을 조사할 것이다.

먼저 함수 $y=f(x)$의 그래프를 만들어 보자. <그림 10-11>에서 A열에는 그래프의 범주 n을 위해 0부터 200까지 입력하였다. 셀 B6에는 x의 초기값을 입력하고, B 열에는 dx크기로 커지는 x 값들을 출력하도록 한다. dx크기는 셀 B1에 입력되어 있다. 여기서는 $dx=0.04$를 사용하고, 초기값은 $x_0=-4$로 하여, 이 함수의 정의역은 $-4 \le x \le 4$가 된다. 셀 C6에 함수 공식(여기에서 이 공식은 $f(x)=0.5x^3-x$)을 입력하고, 열 아래에 그것을 복사해 넣는다.

이제 셀 E1에 평행이동의 크기 c를 정하고, D열 아래에 $x+c$의 계산값이 출력되도록 한다. 그 다음 셀 C6의 함수식을 셀 E6에 복사해서 $f(x+c)$의 값을 산출하도록 하고 E열의 나머지 아래 부분에도 그것을 복사한다. 원래 정의역에 대한 함수 $g(x)=f(x+c)$의 그래프를 얻기 위해, E열을 그

래프로 드래그하여 <그림 10>의 붉은색 곡선을 만든다. 이 과정의 세부사항은 [1]을 참고하라.

그래프에서, 어떻게 원래 함수가 c만큼 평행이동 되었는지를 나타내기 위해 우리는 수평 점선을 첨가하였다. 이 직선의 최적 위치는 우리가 사용한 함수에 따라 변할 것이기 때문에, 우리는 셀 B2에 라벨번호(여기에서는 153)을 입력하였는데, 이것은 153번째 점의 y값에서 수평점선을 만들어낸다. F열과 G열에서 공식은 점선의 끝점을 산출한다.

이제 남은 일은 스크롤 바를 만드는 것과, 그것을 c값을 변하게 만드는 셀과 연결하는 것이다. 하지만, 엑셀에서 스크롤 바는 단지 음이 아닌 정수값만 취할 수 있다. 때문에 셀 F1에 정수를 입력하고, 이 셀과 스크롤바를 연결한 다음, 크기 1을 단위로 0부터 500까지 변화시키기로 한다. 그 다음 셀 E1에 수평이동 값 c를 (250－ F1)/100으로 계산하면 c는 -2.5에서 2.5까지 크기 0,01의 단계로 변하게 된다. 스크롤 바의 슬라이더를 움직이면서, 평행이동을 시각적으로 확인 할 수 있다.

수직이동과 조정비율을 포함하는 예로부터 학생들의 탐구를 동기화 하는 많은 가능성들을 엿볼 수 있다.

	A	B	C	D	E	F	G
1	dx	0.04		c	1.92	58	
2	label	153			c= 1.92		
3							
4							
5	n	x	f(x)	x+c	f(x+c)	xx	yy
6	0	-4.00	-28.00	-2.08	-2.42	2.12	2.64
7	1	-3.96	-27.09	-2.04	-2.20	0.20	2.64
8	2	-3.92	-26.20	-2.00	-2.00		
9	3	-3.88	-25.33	-1.96	-1.80		
10	4	-3.84	-24.47	-1.92	-1.62		
11	5	-3.80	-23.64	-1.88	-1.44		
12	6	-3.76	-22.82	-1.84	-1.27		
13	7	-3.72	-22.02	-1.80	-1.12		
14	8	-3.68	-21.24	-1.76	-0.97		
15	9	-3.64	-20.47	-1.72	-0.82		
16	10	-3.60	-19.73	-1.68	-0.69		
17	11	-3.56	-19.00	-1.64	-0.57		
18	12	-3.52	-18.29	-1.60	-0.45		
19	13	-3.48	-17.59	-1.56	-0.34		
20	14	-3.44	-16.91	-1.52	-0.24		
21	15	-3.40	-16.25	-1.48	-0.14		
22	16	-3.36	-15.61	-1.44	-0.05		
23	17	-3.32	-14.98	-1.40	0.03		



<그림 10>

	A	B	C	D	E	F
1	dx	0.04		c	=(250-F1)/100	58
2	label	153			="c= "&TEXT(E1,"0.00")	
3						
4						
5	n	x	f(x)	x+c	f(x+c)	
6	0	-4	=0.5*B6^3-B6	=B6+$E$1	=0.5*D6^3-D6	
7	=1+A6	=B6+$B$1	=0.5*B7^3-B7	=B7+$E$1	=0.5*D7^3-D7	
8	=1+A7	=B7+$B$1	=0.5*B8^3-B8	=B8+$E$1	=0.5*D8^3-D8	

<그림 11>

학생들은 R^2에서 R^3의 어떤 투시도를 설계함으로써 두 실변수에 대한 3차원 이미지와 함수를 나타내는 그림을 표준 교과서의 절차(scheme)에 따라서 만들어낼 수도 있다. <그림 12>에 한 예가 제시되어 있다. 게다가, 학생들은 복소함수에 대한 정의역과 이미지를 나타내는 R^2의 두 그래프를 사용할 수 있다! <그림 13>은 수평선과 수직선 격자위에 복소 역수 함수 $f(z)=\frac{1}{z}$의 그림을 보여준다([4]). 이 주제는 오늘날의 교육과정에 종종 포함되어 있지 않은 역사적으로 중요한 개념들을 영재 학생들이 탐구할 수 있는 분야의 예를 제공한다.



<그림 12>



<그림 13>

III. 수학적 모델링

아마도 영재 학생들에게 가장 큰 잠재성을 제공하는 스프레드시트 수학의 분야는 수학적 모델링에 있는 것 같다. 이것은 중요한 주제일 뿐 아니라, 왜 수학이 유용하면서 흥미로운지 입증하는 아이디어들의 도입을 위한 방법을 교사들에게 제공한다. 더욱이 수학적 모델링을 하면서, 스프레드시트 모델을 만드는 과정 자체가 학생들로 하여금 새로운 수학을 배우고 발달시키도록 할 수 있다. 이것은 [8]의 예에서 확인 할 수 있다. 유전학, 행성운동, 열의 흐름, 인구 증가, 재고목록, 약의 투약, 입

법부의 분배와 같은 다양한 분야에서 풍부한 수행 주제들을 찾을 수 있다. 또한 스프레드시트는 학생들이 미적분학과 미분방정식과 같은 전통적인 선행필수과목에 대한 배경지식을 몰라도 이런 주제들을 효과적으로 수행할 수 있게 하기도 한다.

유행병 확산에 대한 기본SIR모델을 고려해 보자. 이 모델에서 특정 크기의 고립된 지역을 조사하였다. 거주민들을 세 종류로 구분하였다. S집단은 현재 감염되지는 않았지만, 감염된 사람과 접촉하면 병에 걸리는 사람, I집단은 현재 감염된 사람이다. 이 유행병은 치명적이지는 않고, 감염된 후 일정한 시간 경과 후 치료되면 더 이상 이 병에 걸리지 않는다. 이런 사람들을 R집단 이라 하자.

<그림 14>, <그림 15>에 이 지역 인구 2000명, 처음 감염된 사람(I그룹)은 3명으로 설정하였다. 어느 기간동안-한달이라 하자-주민들은 다른 사람과 접촉을 하게 된다. 그 가능성의 정도를 알 수 있다고 하여 접촉률 p라고 하고, p=0.001이라고 하자. S그룹의 사람이 I그룹의 사람과 접촉을 하면 병에 감염된다. 그러므로 주어진 기간동안 k명의 감염된 사람이 있다면 감염되지 않기 위해서 S 그룹의 사람들은 감염된 사람I k명을 모두 피해야 한다. 따라서 감염되지 않을 확률은 $(1-p)^k$이고 감염될 확률은 $1-(1-p)^k$이다.

이 모델을 만들기 위해서 첫 열에는 기간을 표시하고, C4에는 감염된 사람I의 수 3을 입력하고, 셀 D4에는 R집단의 사람수 0을 입력한다. 그리고 S집단의 사람수를 계산하여 B4에 입력한다. 첫 번째 기간동안 새롭게 감염된 사람I의 수는 E4에 감염된 사람I의 수와 S집단의 사람수의 곱으로 나타낸다. 새롭게 나타난 R 집단의 사람들은 치료율과 현재 감염된 사람의 I 수의 곱이다. 치료율을 1이라고 하고 변수로 하자. - 치료율은 원하는 데로 조정할 수 있다. 다음 열에서 다음 기간의 S집단의 사람 수를 이전의 S집단의 사람 수에서 새롭게 감염된 사람의 I수를 빼서 계산한다. 감염된 사람의 총 수는 이전에 감염된 사람의 수에서 새롭게 감염된 사람의 수를 더하고, 새롭게 치료된 사람의 수를 빼서 계산한다. R집단의 사람 수는 이전의 사람 수에 새로운 사람 수를 추가해서 더하면 된다. 각각의 계산을 수행하면 우리가 원하는 모든 결과를 각 열에서 볼 수 있다. 아래 <그림 14>에서 치료율과 접촉율은 절대참조이다.

	A	B	C	D	E	F
1	popn	2000	cnt rate	0.001	cure rate	1
2						
3	period	susceptible	infected	removals	new infect	new remove
4	0	=B1-C4	3	0	=B4*(1-(1-$D$1)^C4)	=C4*$F$1
5	=1+A4	=B4-E4	=C4+E4-F4	=D4+F4	=B5*(1-(1-$D$1)^C5)	=C5*$F$1
6	=1+A5	=B5-E5	=C5+E5-F5	=D5+F5	=B6*(1-(1-$D$1)^C6)	=C6*$F$1

<그림 14>

아래 <그림 15>에서 계산결과를 볼 수 있다.

	A	B	C	D	E	F
1	popn	2000	cnt rate	0.001	cure rate	1
2						
3	period	susceptible	infected	removals	new infect	new remove
4	0	1997.0	3.0	0.0	6.0	3.0
5	1	1991.0	6.0	3.0	11.9	6.0
6	2	1979.1	11.9	9.0	23.4	11.9
7	3	1955.7	23.4	20.9	45.3	23.4

<그림 15>

그래프를 사용하면 이 모델을 더 잘 이해할 수 있다. <그림 16>의 그래프를 보자. 이 모델에서 전염병은 모든 사람이 감염되기 전에 점진적으로 사라진다. 이 모델의 인구, 치료율, 초기 감염자의 수, 접촉율을 조정함으로써 더 여러 가지 경우의 상황을 조사할 수 있다.

이와 같은 모델을 사용하는 것이 영재학생들에게 좋은 과제가 되는 이유는 학생들에게 이 모델이 실재의 경우에 적용되지 않는 이유를 생각하게 해주기 때문이다. 이 문제를 생각하면서 학생들은 몇 가지 사실을 추가하고 싶어 질 것이다. 이를테면 이 지역에 새로운 이주민이 들어오거나, 새로운 아이들이 출생하거나, 죽는 사람이 생기거나, 이 병에 대한 면역력이 단기간동안에만 지속될 수도 있다. 이러한 요소들을 생각하다 보면 이 모델을 여러 가지로 개선시킬 수 있고 더욱 도전적인 과제로 확장 할 수 있다. <그림 17>은 면역력이 12개월 동안만 지속되고, 월간 치료율이 85%인 경우를 나타낸 것이다. 이 경우는 약 2.5년을 주기로 하고 있다는 것을 볼 수 있다.



<그림 16>



<그림 17>

IV. 수학적 알고리즘

스프레드쉬트는 수학적 알고리즘을 수행하기 위한 유용한 도구이다.([5], [7], [8]). 또한 학생들이 그래프적 표현을 창조하게 함으로써 시각적 통찰를 가질 수 있게 한다. 스프레드쉬트의 또 다른 장점은 우리가 손으로 하는 것과 같은 방법으로 작동하므로, 스프레드쉬트 모델을 만드는 것 자체가 알고리즘과 관련된 수학적 능력을 강화시킨다. 뿐만 아니라 일단 스프레드쉬트 모델을 만들면 학생들은 변수를 조정하거나, 초기조건을 조정하거나, 함수를 조정했을 때의 결과를 보면서 이 모델을 더 잘 이해할 수 있다. 대수학, 미적분학, 수치해석, 통계와 같은 분야에는 흥미있는 예들이 풍부하다. 학생들은 열심히 책을 찾거나 인터넷을 조사하여 중요하고 흥미있는 주제를 찾아 계획을 세워 이러한 것을 수행할 수 있다. 대부분의 예는 수행할 수 있는 여러 가지 다른 방법이 있어서 학생들에게 자신의 창조적인 기술을 발전시킬 무제한적인 기회를 제공한다.

이제 우리는 그러한 알고리즘의 한 예로서 연속 실가 함수의 x절편을 찾는 secant method 에 대해 알아보자. 이 알고리즘은 미적분학의 뉴턴 방법(뉴턴 방법은 그 자체로 훌륭한 연구의 주제이다)과 비슷하지만 secant method는 미적분학이나 미분계수 등에 대한 지식을 필요로 하지 않는다.

<그림 18>의 그래프를 참고하여 secant method를 설명할 수 있다. 우선 x절편의 추정값, x_0, x_1를 잡는다. 그리고 두 점 $(x_0, f(x_0))$, $(x_1, f(x_1))$를 지나는 직선을 생각하자. 그 직선이 x축과 만나는 x^2를 x절편으로 추정할 수 있는데 이것은 좋은 추정값은 아니다. 그래서 x_1, x_2를 사용하여 위의 과정을 반복한다. 우리는 원하는 정도의 정확도를 얻을 때까지 이 작업을 계속할 수 있다.



<그림 18>



<그림 19>

이 알고리즘을 적용하기 위해 우선 두 점을 지나는 직선의 기울기 $m=(y-y_0)/(x_1-x_0)$에 주목하자. 이 직선의 방정식은 $y-y_1=m(x-x_1)$이다. y에 0을 대입하고 x에 x_2를 대입하면 $-y_1=m(x_2-x_1)$를 얻고, 따라서 $x_2=x_1-y_1/m$이다.

$f(x) = 0.4x^3 - 5x + 2$인 경우에 이 알고리즘을 구현한 스프레드시트 예제가 <그림 20-21>에 있다. 우선 횟수를 세기 위해서 첫째 열을 사용한다. B열에는 초기 추정치로서 0과 1을 차례로 입력하고 C열에서는 $y = f(x)$를 계산한다. 마지막으로 D열에서는 마지막 두 점의 좌표를 사용하여 기울기 m을 계산한다. 셀B4에는 위에서 유도한 x_2 공식을 입력한다. 그런 후에 이 공식들의 각각을 드래그하여 열을 따라 복사한다. 이 때 모든 칸의 변수는 상대참조이다. 이 경우에는 secant method 알고리즘이 근접한 x절편으로 빠르게 수렴하는 것을 확인할 수 있다. 그러나 뉴턴 방법[8]을 이용한 <그림 22>의 경우에서 보듯이 항상 빠르게 수렴하는 것은 아니다. 쌍방향의 모델을 디자인할 때에는 대략 x절편이 어디에 위치하는지 알기 위해 위 함수의 그래프를 그려 보는 것이 도움이 된다.

	A	B	C	D
1	n	x	y	slope
2	0	0	2	
3	1	1	-2.6	-4.6
4	2	0.434783	-0.14104	-4.35047
5	3	0.402364	0.014238	-4.78965
6	4	0.405336	-4.4E-05	-4.80429
7	5	0.405327	-1.3E-08	-4.80285
8	6	0.405327	1.18E-14	-4.80285

<그림 20>

	A	B	C	D
1	n	x	y	slope
2	0	0	=0.4*B2^3-5*B2+2	
3	=1+A2	1	=0.4*B3^3-5*B3+2	=(C3-C2)/(B3-B2)
4	=1+A3	=B3-C3/D3	=0.4*B4^3-5*B4+2	=(C4-C3)/(B4-B3)
5	=1+A4	=B4-C4/D4	=0.4*B5^3-5*B5+2	=(C5-C4)/(B5-B4)
6	=1+A5	=B5-C5/D5	=0.4*B6^3-5*B6+2	=(C6-C5)/(B6-B5)
7	=1+A6	=B6-C6/D6	=0.4*B7^3-5*B7+2	=(C7-C6)/(B7-B6)
8	=1+A7	=B7-C7/D7	=0.4*B8^3-5*B8+2	=(C8-C7)/(B8-B7)

<그림 21>

이 같은 모델을 통해 학생들은 초기 추정값 x_0, x_1을 여러 가지로 바꾸어 시험해 봄으로써 패턴을 발견할 수 있다. 또 때에 따라선 이 알고리즘이 항상 수렴하지 않고 진동할 수도 있기 때문에 어떤 재미있는 결과를 얻을 수 있을 것이다. 이 알고리즘은 <그림 22>의 뉴턴 방법에서처럼 우리가 처음 추정한 값과 먼 거리에 있는 예상치 못한 값으로 수렴할 수도 있다. 때로는 연역적 탐구를 통해 알고리즘이 그렇게 작용하는 이유를 밝힐 수 있다. 뿐만 아니라 학생들은 뉴턴 방법, false position, 이분법, 고정점 방법 등의 x절편을 찾는 다른 방법들에 관심을 갖질 수도 있을 것이다.



<그림 22>

사실 수치해석적인 알고리즘의 대부분([5],[7])은 스프레드시트를 사용하는 교육의 훌륭한 소재가 된다. 그 주제들은 보통 기초적인 미적분학 지식을 필요로 하는데 수치적분, 수치미분, 멱급수, 미분방정식의 수치해, 연립일차방정식의 반복해, 고유값 등을 포함한다.

V. 추가할 수 있는 주제들

위에서 언급된 예에 덧붙여 이 절에서는 영재들의 생산적으로 탐구할 수 있는 몇 가지 그래프를 제시하고자 한다. 예를 들어, 극한의 $\varepsilon-\delta$정의(<그림 23>)나 곡선 아래 넓이의 사각형 넓이를 이용한 근사(<그림 24>)를 역동적인 애니메이션으로 디자인해 볼 수 있다.



<그림 23>



<그림 24>

게다가 예술적 기질이 있는 학생들은 <그림 25-27>([4])과 같이 스스로 창작한 문양은 물론 전통적이고 민족적인 문양도 디자인해 볼 수 있을 것이다.



<그림 25>



<그림 26>



<그림 27>

스프레드시트는 통계, 예술, 미적분, 미분방정식, 디자인, 수치해석, 3차원 그래프, 경영, 경제, 과학, 생태학, 환경학, 그래픽 디자인, 사회학, 레크리에이션, 프랙탈 등의 분야에 광범위하게 응용된다. 이 중 몇 가지에 대해 참고할 수 있는 자료들을 참고문헌에 소개한다.

참 고 문 헌

Arganbright, Deane (2005). Enhancing Mathematical Graphical Displays in Excel through Animation, *Spreadsheets in Education* **2(1)**, pp.125-147. http://www.sie.bond.edu.au/classresource.asp?id=10

Arganbright, Deane (2005). Developing Mathematics Creativity with Spreadsheets, *Journal of the Korea Society of Mathematical Education, Series D: Research in Mathematical Education* **9(3)**, Sept. pp.187-201.

Arganbright, Deane (2002). Creative Spreadsheet Graphics in Mathematics Teaching and Modeling, *Technology in Mathematics Teaching*, Schriftenreihe Didaktik der Mathematik, **26**, Oesterreichischer Bundesverlag, Vienna, pp.153-156.

Arganbright, Deane (1993). *Practical Handbook of Spreadsheet Curves and Geometric Constructions*, CRC Press, Boca Raton, Florida.

Arganbright, Deane (1985). *Mathematical Applications of Electronic Spreadsheets*, McGraw-Hill, New York.

Hill, David and Lila Roberts. *Demos with Positive Impact* (Web site): http://mathdemos.gcsu.edu/morelinks2.html

McLaren, David (1997). *Spreadsheets and Numerical Analysis,* Chartwell-Bratt.

Neuwirth, Erich and Deane Arganbright (2004). *The Active Modeler: Mathematical Modeling with Microsoft Excel*, Brooks-Cole.

Neuwirth, Erich. *Spreadsheets, Mathematics, Science, and Statistics Education* (Web site): http://sunsite.univie.ac.at/Spreadsite/

Sugden, Steve, ed. *Spreadsheets in Education.* http://www.sie.bond.edu.ac/